Di3590l1.doc

*******************************************************

*        µC protects DC Motor                     *

*******************************************************

$nolist

*uC908QT2 Memory Addresses

prtA     EQU $00 ; prtA

ddrA     EQU $04 ; direction prtA register

config2  EQU $1E

config1  EQU $1F

TSC      EQU $20 ;Timer Status & Control reg.

TSC0     EQU $25 ;TIM Channel 0 Status and Control Register

TSC1     EQU $28 ;TIM Channel 1 Status and Control Register

TOF      EQU 7   ;TOF flag

TSTOP    EQU 5   ;Timer Stop bit

TmodH    EQU $23 ;Timer modulo register, HIGH

TmodL    EQU $24 ;Timer modulo register, LOW

ADSCR    EQU $3C ;A/D Satus & Control reg

COCO     EQU 7   ;A/D Conversion Complete flag

ADR      EQU $3E ;A/D data register

RAM      EQU $0080

ROM      EQU $F800

$list

******************************************************

*I/O prtA

Input       EQU  4     ;pA4 read ADC proportional to motor current

Out         EQU  5     ;pA5 output to control MOSFET motor driver

*CONSTANTS

PERIOD_PWM  EQU $1F4   ;period PWM=10ms (10*50=500=$1F4)

DELAY       EQU 100T   ;Delay number = DelayTime/PWM period=1s/10ms=100

Vmax        EQU $2B    ;Max.threshold value=0.85V [256*0.85/5=43.52=$2B]

Vmin        EQU $0F    ;Min.threshold value=0.3V  [256*0.3/5=15.36=$0F]

MotorF      EQU 2      ;Motor flag: 0- make delay, 1- skip delay

*VARIABLES

          ORG   RAM

flag        RMB   1    ;Flag register

cnt         RMB   1    ;Counter of PWM periods

*INITIALIZATION

            ORG ROM

init        RSP                   ;$FF-->SP

            CLRA

            CLRH                  ;clear H:X

            CLRX

            CLR   flag            ;0 --> MotorF

            CLR   cnt             ;0 --> cnt

            MOV   #$01,CONFIG1    ;COP disabled

            LDA   $FFC0           ;adjusting frequency

            STA   $0038

            MOV   #%00100000,prtA ;Start motor

            MOV   #%00100000,DDRA ;set I/O prtA

            LDHX  #PERIOD_PWM     ;PWM PERIOD-->TmodH

            STHX  TmodH

            MOV   #%00010110,TSC  ;clear & start TIMER,prescaler:64

***************************************************************

MAIN        BRCLR TOF,TSC,*       ;wait for the end of PERIOD_PWM

            BCLR  TOF,TSC         ;TOF reset

            BRSET MotorF,flag,Run ;if MotorF=1, then skip delay and run motor

            inc   cnt             ;make DELAY = 2.5 sec (2500ms/10ms=250 TOF)

            lda   cnt

            cmp   #250T

            bls   Main

Run         mov   #%00000010,ADSCR;start ADC2 at pA4

            BRCLR COCO,ADSCR,*    ;wait for end of ADC conversion 

            LDA   ADR             ;ADC data ->Acc

            CMP   #Vmin           ;if Voltage < Vmin, then stop motor

            BLS   fault

            CMP   #Vmax           ;if Voltage > Vmax, then stop motor

            BHS   fault

            BSET  out,prtA        ;Run motor

            BSET  MotorF,flag     ;Set MotorF=1

            BRA   MAIN

fault       CLR   prtA            ;Stop motor

            CLR   TSC0

            CLR   TSC1

lp          BRA   lp              ;Infinity loop

**************************************************************

            ORG $FFFE             ;Vector reset

            FDB init              ;Set start address

.end

.nolist

